

Llamadas a Función:

Ejemplo función no-hoja (llama a otra)

```
void absVect(int vector[], int len) {  
    for (int i=0; i<len; i++)  
        v[i] = abs(v[i]);  
}
```

```
absVect:  
    push {r4-r6, lr} // salvamos registros  
                        // salvados que usamos  
    mov r4, r0        // pasamos parámetros  
    mov r5, r1        // a reg. salvados  
    mov r6, #0        // para no perderlos  
loop:  
    cmp r6, r5  
    bge exit  
    ldr r0, [r4]      // cargamos parámetro  
                        // de llamada hija  
    bl abs            // llamada hija  
    str r0, [r4], #4  // usamos resultado  
    add r6, r6, #1  
    b loop  
exit:  
    pop {r4-r6, lr}  // restauramos regs.  
                        // salvados  
    bx lr            // retornamos
```

1. Cuidado con los registros no salvados que los perdemos en la llamada a función hija (**perdemos parámetros de entrada**: vector, len)
2. Salvamos en PILA los registros salvados que vayamos a utilizar (**LR siempre** porque se pierde en la llamada BL)
3. Usamos los parámetros de entrada y registros salvados como queramos
4. Escribimos **parámetros de llamada** (r0)
5. Llamamos a la función hija
6. Utilizamos el resultado como necesitemos (r0)
7. Escribimos resultado (si lo hay) en r0
8. Restauramos los registros salvados que hemos utilizado (al menos LR)
9. Retornamos de la función